
Пример 1. Създайте приложение, съдържащо два командни бутона и един
етикет. За свойство Caption на етикета задайте стойността “Hello”, а за двата
бутона – съответно “On” и “Off”. При позициониране на показалеца на мишката
върху един от двата бутона той да се променя съответно във вида 1-Arrow, а при
позициониране върху формата – вида на показалеца на мишката да бъде 2- Cross.
Съхранете проекта под име project1.vbp, име на формата-form1.frm.

Пример 3. Създайте приложение за пресмятане на квадратния корен на зададено
произволно число. При въвеждане на даденото число в текстово поле, с помощта на
команден бутон да се пресмята квадратния корен на числото и получения резултат да се
визуализира в етикет. Задайте име на създаденото от вас приложение-Primer3.vbp,
име на формата – frmPrimer3.frm.

Програмен код:
Private Sub cmdExit_Click()

End
End Sub
Private Sub cmdNew_Click()

txtInput.Text = ""
lblResult.Caption = ""
txtInput.SetFocus

End Sub
Private Sub cmdRes_Click()

lblResult.Caption = Sqr(txtInput.Text)
End Sub

Допълнение:При опит за въвеждане на отрицателно число в текстовото поле
да се получава съобщение за грешка.

Private Sub cmdRes_Click()
If txtInput.Text < 0 Then
MsgBox "Въвели сте число, което попада извън дефиниционната област!"
Exit Sub

Else: lblResult.Caption = Sqr(txtInput.Text)
End If

End Sub

Пример 4. Създайте приложение, което с помощта на два командни бутона
увеличава и намалява размерите на дадена картинка(действието наподобява пускане
и вдигане на щора). Приложението съдържа още два командни бутона, с помощта
на които се извършва съответно промяна на картинката и затваряне на формата.
Задайте име на проекта – shtora.vbp, име на форма – shtora.frm.

Dim n As Integer
Dim s, p As Integer

Private Sub cmdChange_Click()
If p = 0 Then
Picture1.Picture = Picture2.Picture
p = 1
ElseIf p = 1 Then
Picture1.Picture = Picture3.Picture
p = 2
ElseIf p = 2 Then
Picture1.Picture = Picture4.Picture
p = 3

Else: Picture1.Picture = Picture5.Picture
p = 0

End If
End Sub

Private Sub cmdClose_Click()
End

End Sub

Private Sub cmdDown_Click()
Do While s < n
Picture1.Height = Picture1.Height + 1
s = s + 1
Loop
End Sub

Private Sub cmdUp_Click()
n = Picture1.Height
s = n
Do While s > 50
Picture1.Height = Picture1.Height - 1
s = s - 1
Loop

End Sub

Програмен код:

Програмен код:
Private Sub cmdAdd_Click()

lstClientList.AddItem txtInput.Text
txtInput.Text = ""
txtInput.SetFocus

End Sub

Private Sub cmdDelete_Click()
If lstClientList.ListIndex >= 0 Then
lstClientList.RemoveItem lstClientList.ListIndex

Else
Beep

End If
End Sub

Private Sub cmdClear_Click()
lstClientList.Clear

End Sub

Private Sub cmdClose_Click()
End

End Sub

Пример 6. Създайте приложение, съдържащо следните елементи: списъчна кутия,
етикет и текстово поле. При избор на даден елемент от списъчната кутия
стойността му да бъде показана и в текстовото поле.
Задайте име на проекта – Fruits.vbp, име на форма – Fruits.frm.

Програмен код:
Private Sub lstFruits_Click()
txtSelection.Text =lstFruits.List(lstFruits.ListIndex)

End Sub

Пример 7. Създайте приложение, което съдържа етикет, текстово поле, три опционални
бутона(Option buttons) и две полета за отметка (Checkbox). С помощта на опционалните
бутони да се извършва избор на шрифт за текста, въведен в текстовото поле (свойство
FontName), както и настройка цвета на фона (свойство BackColor). Посредством полетата за
отметка да се избира стойност(True или False) за свойствата FontBold и FontItalic на
текстовото поле. Задайте име на проекта - TextFormats.vbp, име на форма - TextFormats.frm.

Програмен код:

Private Sub optMSSansSerif_Click ()
txtInput.FontName = "MS Sans Serif"
txtInput.BackColor = vbGreen

End Sub

Private Sub optTimesNewRoman_Click ()
txtInput.FontName = "Times New Roman"
txtInput.BackColor = vbRed

End Sub

Private Sub optArial_Click ()
txtInput.FontName = "Arial"
txtInput.BackColor = vbBlue

End Sub

Private Sub chkBold_Click ()
If chkBold.Value = 1 Then

txtInput.FontBold = True
Else

txtInput.FontBold = False
End If

End Sub

Private Sub chkItalic_Click ()
If chkItalic.Value = 1 Then

txtInput.FontItalic = True
Else

txtInput.FontItalic = False
End If

End Sub

Програмен код:
Option Explicit
Dim original As Single

Private Sub Form_Load()
imglight.Picture = imgYellow.Picture

End Sub

Private Sub cmdChange_Click()
Select Case imglight.Picture

Case imgGreen.Picture
imglight.Picture = imgYellow.Picture

Case imgYellow.Picture
imglight.Picture = imgRed.Picture

Case imgRed.Picture
imglight.Picture = imgGreen.Picture

original = imgCar.Left
Do While imgCar.Left > 0

imgCar.Left = imgCar.Left - 1
Loop
imgCar.Left = original

End Select
End Sub

Пример 10. Създайте приложение, което съдържа два етикета и един
Timer control, показващо текущата дата и текущия час, обновявайки го всяка
секунда. Задайте име на проекта – Clock.vbp, име на форма – Clock.frm.

Програмен код:
Private Sub Form_Load()

lblDate.Caption = Date
End Sub
Private Sub Timer1_Timer()

lblTime.Caption = Time
End Sub

Програмен код:
Option Explicit
Dim original As Single
Dim f As Integer

Private Sub cmdStartLights_Click()
tmrLights.Enabled = TRUE

End Sub

Private Sub cmdStopLights_Click()
tmrLights.Enabled = FALSE

End Sub

Private Sub Form_Load()
imglight.Picture = imgYellow.Picture
f = 0

End Sub

Private Sub tmrLights_Timer()
Select Case imglight.Picture

Case imgGreen.Picture
imglight.Picture = imgYellow.Picture

Case imgYellow.Picture
If f = 0 Then imglight.Picture = imgRed.Picture
If f = 1 Then
imglight.Picture = imgGreen.Picture
original = imgCar.Left
Do While imgCar.Left > 0
imgCar.Left = imgCar.Left - 1

Loop
imgCar.Left = original

f = 0
End If

Case imgRed.Picture
imglight.Picture = imgYellow.Picture
f = 1

End Select
End Sub

Пример 12.Анимация, Messagebox, Picture box.
Създайте приложение, съдържащо следните елементи: PictureBox, три командни бутона,
две картинки (Image) и един етикет. Елемента Picturebox се явява контейнер за двата
командни бутона – Fly и GoBack и за картинката, визуализираща самолет (свойство
Picture=plane.ico).
Забележка: Тъй като PictureBox е контейнер, то е необходимо той да бъде нарисуван във
формата преди другите три елемента(контрола), съдържащи се в него.

При кликване върху картинката, под която има надпис Click Me! се появява следното
съобщение:

Непосредствено преди опит за изход от приложението с помощта на бутон Quit да бъде
визуализирано следното предупредително съобщение:

Съхранете проекта под име Flying.vbp, а формата с име frmFlying.frm.

Програмен код:
Private Sub cmdFly_Click()

Do While imgplane.Left < picflying.Width
imgplane.Left = imgplane.Left + 3
imgplane.Top = imgplane.Top + 1

Loop
End Sub
Private Sub cmdGoBack_Click()

Do While imgplane.Left > 0
imgplane.Left = imgplane.Left - 3
imgplane.Top = imgplane.Top - 1

Loop
End Sub
Private Sub imgme_Click()

MsgBox "Thanks for clicking on me!", vbInformation, "Info Message"
End Sub
Private Sub Quit_Click()

If MsgBox("Do you really want to quit?", vbQuestion + vbYesNo +_
vbDefaultButton2, "Flying") = vbYes Then

End
End If

End Sub

Пример 13. Създайте приложение, което съдържа следните елементи: етикет, две рамки
(Frames), във всяка от които има група от три опционални бутона(Option buttons).При избор
на даден опционален бутон от рамката ForeGround Color се извършва настройка на
свойството ForeColor на етикета. При избор на даден опционален бутон от рамката
BackGround Color се извършва настройка на свойството BackColor на етикета. Задайте име
на проекта – Hello.vbp, име на форма – Hello.frm.

Програмен код:

Private Sub optForeground_Click(Index As Integer)
Select Case Index

Case 0: lblHello.ForeColor = vbRed
Case 1: lblHello.ForeColor = vbYellow
Case 2: lblHello.ForeColor = vbBlue

End Select
End Sub

Private Sub optBackground_Click(Index As Integer)
Select Case Index

Case 0: lblHello.BackColor = vbBlack
Case 1: lblHello.BackColor = vbWhite
Case 2: lblHello.BackColor = vbGreen

End Select
End Sub

Пример 14. Stopwath Application
Създайте приложение, което съдържа три командни бутона и шест етикета. При
натискане на бутони StartTiming и EndTiming в етикети се отбелязва съответно
времето, в което е бил натиснат всеки един от тези бутони. На базата на
стойностите в тези два етикета се получава стойността, която се записва в
третия етикет. Тази стойност представлява продължителността във
времето(ElapsedTime), започваща с натискането на бутона StartTiming и
завършваща с натискането на бутона EndTiming. Бутон Exit служи за изход от
приложението. Задайте име на проекта–Stopwatch.vbp, име на формата-
Stopwatch.frm.

Програмен код:
Option Explicit
Dim StartTime as Variant
Dim EndTime as Variant
Dim ElapsedTime as Variant

Private Sub cmdStart_click()
StartTime=Now
lblStart.Caption=Format(StartTime, “hh:mm:ss”)
lblEnd.Caption=””
lblElapsed.Caption=””

End sub

Private Sub cmdEnd_click()
EndTime=Now
ElapsedTime=EndTime-StartTime
lblEnd.Caption=Format(EndTime, “hh:mm:ss”)
lblElapsed.Caption=Format(ElapsedTime, “hh:mm:ss”)

End sub

Private Sub cmdExit_click()
end

end sub

Пример 15. Създайте приложение, показващо текущия месец, ден и година.
Също така и текущото време, обновявайки го всяка секунда. Прозореца би
трябвало да наподобява страница на календар. Задайте име на проекта –
Calendar.vbp, име на форма – Calendar.frm.

lblDay

lblTime

timDisplay

lblMonth

lblNumber

lblYear

Програмен код:
Private sub timDisplay_Timer()

Dim Today As Variant
Today=Now
lblDay.Caption=Format(Today, “dddd”)
lblMonth.Caption=Format(Today, “mmmm”)
lblYear.Caption=Format(Today, “yyyy”)
lblNumber.Caption=Format(Today, “d”)
lblTime.Caption=Format(Today, “h:mm:ss ampm”)

end sub

Пример 16. Image Viewer
Създайте приложение, целящо преглеждане на графични файлове (с
разширение *.bmp; *.wmf; *.ico). При избор на конкретно устройство,
директория и име на файл да се визуализира селектирания файл в елемент
Image. Във формата е включен и етикет, който ще показва пълния път за
достъп до селектирания файл. Задайте име на проекта–ImgViewer.vbp,
име на формата-ImgViewer.frm.

Програмен код:
Private Sub Drive1_Change()

Dir1.Path = Drive1.Drive
End Sub

Private Sub Dir1_Change()
File1.Path = Dir1.Path

End Sub

Private Sub File1_Click()
Dim Fullpath As String
If Right(File1.Path, 1) = "\" Then

Fullpath = File1.Path & File1.List(File1.ListIndex)
Else

Fullpath = File1.Path & "\" & File1.List(File1.ListIndex)
End If
lblFullpath.Caption = Fullpath
imgDisplay.Picture = LoadPicture(Fullpath)

End Sub

Пример 17.Създайте приложение – калкулатор:

Задайте име на проекта–Calculator.vbp, име на формата-Calculator.frm.

Програмен код:
Dim result, a(1) As Single
Dim ind As Integer
Dim t As String
Private Sub Command1_Click(Index As Integer)

Label1.Caption = Label1.Caption +
Command1(Index).Caption
End Sub
Private Sub Command2_Click()

Label1.Caption = Label1.Caption + Command2.Caption
Command2.Enabled = False

End Sub
Private Sub Command3_Click()

a(ind) = Val(Label1.Caption)
Select Case t

Case "+"
result = a(0) + a(1)

Case "-"
result = a(0) - a(1)

Case "*"
result = a(0) * a(1)

Case "/"
If a(1) = 0 Then

MsgBox "Деление на 0!"

Else
result = a(0) / a(1)

End If
End Select
Label1.Caption = result

End Sub
Private Sub Command4_Click(Index As Integer)

t = Command4(Index).Caption
Command2.Enabled = True
a(ind) = Val(Label1.Caption)
Label1.Caption = ""
ind = 1

End Sub
Private Sub Command5_Click()

Label1.Caption = ""
ind = 0
Command2.Enabled = True

End Sub
Private Sub Form_Load()

Label1.Caption = ""
ind = 0

End Sub

Програмен код:
Option Explicit
Dim Months(12) As String
Dim Days(12) As Integer
Dim Grade As String
Private Sub cmdExit_Click()

End
End Sub
Private Sub cmdLoad_Click()

cdlbox.ShowOpen
imgStudent.Picture = LoadPicture(cdlbox.filename)

End Sub
Private Sub cmdNew_Click()

txtName.Text = ""
imgStudent.Picture = LoadPicture("")

End Sub
Private Sub cmdShow_Click()

Dim Is_Leap As Integer
Dim Msg As String, Age As Integer, Pronoun As String
Dim M As Integer, D As Integer, Y As Integer

If vsbMonth.Value = 2 And ((vsbYear.Value Mod 4 = 0 And_
vsbYear.Value Mod 100 <> 0) Or vsbYear.Value Mod 400 = 0) Then

Is_Leap = 1
Else

Is_Leap = 0
End If
If vsbDay.Value > Days(vsbMonth.Value) + Is_Leap Then
MsgBox "Only" + Str(Days(vsbMonth.Value) + Is_Leap) + " days in " +_

Months(vsbMonth.Value), vbOKOnly + vbCritical, "Invalid Birth Date"
Exit Sub

End If
M = Val(Format(Now, "mm"))
D = Val(Format(Now, "dd"))
Y = Val(Format(Now, "yyyy"))
Age = Y - vsbYear
If vsbMonth.Value > M Or (vsbMonth.Value = M And vsbDay > D) Then Age = Age - 1

If Age < 0 Then
MsgBox "Birth date is before current date.", vbOKOnly +_

vbCritical, "Invalid Birth Date"
Exit Sub

End If
If txtName.Text = "" Then
MsgBox "The profile requires a name.", vbOKOnly + vbCritical,_

"No Name Entered"
Exit Sub

End If
Msg = txtName.Text + " is a student in the " + Grade + " grade." + vbCr
If optSex(0).Value = True Then Pronoun = "He " Else Pronoun = "She "
Msg = Msg + Pronoun + " is" + Str(Age) + " years old." + vbCr
MsgBox Msg, vbOKOnly, "Student Profile"
End Sub

Private Sub Form_Load()
Months(1) = "January": Days(1) = 31
Months(2) = "February": Days(2) = 28
Months(3) = "March": Days(3) = 31
Months(4) = "April": Days(4) = 30
Months(5) = "May": Days(5) = 31
Months(6) = "June": Days(6) = 30
Months(7) = "July": Days(7) = 31
Months(8) = "August": Days(8) = 31
Months(9) = "September": Days(9) = 30
Months(10) = "October": Days(10) = 31
Months(11) = "November": Days(11) = 30
Months(12) = "December": Days(12) = 31
lblMonth.Caption = Months(vsbMonth.Value)
lblDay.Caption = Str(vsbDay.Value)
lblYear.Caption = Str(vsbYear.Value)
Grade = "first"
End Sub
Private Sub optLevel_Click(Index As Integer)
Select Case Index
Case 0

Grade = "first"

Case 1
Grade = "second"

Case 2
Grade = "third"

Case 3
Grade = "fourth"

Case 4
Grade = "fifth"

Case 5
Grade = "sixth"

End Select
End Sub
Private Sub vsbDay_Change()

lblDay.Caption = Str(vsbDay.Value)
End Sub
Private Sub vsbMonth_Change()

lblMonth.Caption = Months(vsbMonth.Value)
End Sub
Private Sub vsbYear_Change()

lblYear.Caption = Str(vsbYear.Value)
End Sub

Пример 19. Visual Basic I Access.
Създайте приложение, съдържащо три текстови полета, етикет и един контрол Data.
Използваме готова база данни Employees.mdb, съдържаща таблица Employee Details с
три полета: FirstName, SurName и Salary. При преглеждане на различните записи в
базата данни с помощта на контрола Data в съответните текстови полета се показват
данните за всеки запис от таблицата datEmployees. Задайте име на проекта–
Datacontrol.vbp, име на формата-frmDatacontrol.frm.

Програмен код:
Private Sub datEmployees_Reposition()

Select Case txtSalary.Text
Case Is > 40000: lblMessage.Caption = "seriously rich"
Case 30000 To 39999: lblMessage.Caption = "almost rich"
Case Else: lblMessage.Caption = "has far to go"

End Select
End Sub

Пример 20. Работа с готова база данни.
На базата на пример 19 да се добавят към приложението бутони,
осъществяващи действията – добавяне, изтриване, търсене и промяна на
записи в база данни.

Програмен код:

Private Sub cmdAdd_Click()
datEmployees.Recordset.MoveLast
datEmployees.Recordset.AddNew
datEmployees.Recordset![FirstName] = InputBox("Нови данни: Въведи малко име: ")
datEmployees.Recordset![Surname] = InputBox("Нови данни: Въведи фамилия: ")
datEmployees.Recordset![Salary] = InputBox("Нови данни: Въведи заплата: ")
datEmployees.Recordset.Update

End Sub

Private Sub cmdAmend_Click()
Dim thename, thesearch As String
datEmployees.Recordset.MoveFirst
thename = InputBox("Who are looking for?")
thesearch = "[Firstname]= '" & thename & "'"
datEmployees.Recordset.FindFirst thesearch
If Not datEmployees.Recordset.NoMatch Then

datEmployees.Recordset.Edit
datEmployees.Recordset![Salary] = InputBox("Enter new salary")
datEmployees.Recordset.Update

End If
End Sub

Private Sub cmdDelete_Click()
datEmployees.Recordset.Delete
datEmployees.Recordset.MoveNext

End Sub

Private Sub cmdSearch_Click()
Dim name As String
Dim searchName As String
name = InputBox("Enter First Name to search")
searchName = "[Firstname]= '" & name & "'"
datEmployees.Recordset.MoveFirst
datEmployees.Recordset.FindFirst searchName
If datEmployees.Recordset.NoMatch Then
MsgBox "Couldn't find this name"

End If
End Sub

Private Sub datEmployees_Reposition()
Select Case txtSalary.Text
Case Is > 40000: lblmessage.Caption = "seriously rich"
Case 30000 To 39999: lblmessage.Caption = "almost rich"
Case Else: lblmessage.Caption = "has far to go"

End Select
End Sub

Решение:

Втора стъпка:

Програмен код:

Private Sub DBCombo1_Click(Area As Integer)
If Area = dbcAreaList Then

List1.AddItem DBCombo1.Text
End If

End Sub

Private Sub DBList1_Click()
List2.AddItem DBList1.Text

End Sub

Private Sub DBGrid1_MouseUp(Button As Integer, Shift As Integer,_
X As Single, Y As Single)
Dim theCol As Integer, Bookmark As Variant
theCol = DBGrid1.ColContaining(X)
Bookmark=DBGrid1.RowBookmark(DBGrid1.RowContaining(Y))
List3.AddItem DBGrid1.Columns(thecol).CellValue(Bookmark)

End Sub

Допълнение:Аналогично на пример 20 могат да бъдат добавени бутони към
формата, реализиращи действията – добавяне, изтриване, търсене и промяна
на записи в базата данни.

Пример 22. Пример за работа с меню.
Създайте приложение, при което се използва вече създадената форма
frmprimer1.frm(от пример 3) и към нея се реализира хоризонтално (а след това
и вертикално) меню, съдържащо командите – Нови данни, Пресмятане на
квадратен корен и Изход.

Указание: Програмният код, съответстващ на процедури click за трите
командни бутона се записва като програмен код за процедурите click, само че
отнасящи се за командите от менюто.
Менюто за приложението се състои от три команди:

� Нови данни - mnuNew
� Пресмятане на квадратен корен - mnuRes
� Изход - mnuExit

Програмен код:

Private Sub mnuExit_Click()
End

End Sub

Private Sub mnuNew_Click()
txtInput.Text = ""
lblResult.Caption = ""
txtInput.SetFocus

End Sub

Private Sub mnures_Click()
If txtInput.Text < 0 Then
MsgBox "Въвели сте число, което попада извън дефиниционната област!"
Exit Sub

Else: lblResult.Caption = Sqr(txtInput.Text)
End If

End Sub

Пример 23. Работа с менюта, Common Dialog control.
Създайте приложение, което съдържа следните елементи на ГПИ:
RichTextBox, CommonDialog. Към проекта реализирайте меню, командите за
което са описани по-нататък. Приложението има за цел да бъдат реализирани
действията, свързани с преглеждане, форматиране и съхраняване на файлове.
Задайте име на проекта–FileViewer.vbp, име на формата-FileViewer.frm.



СЪБИТИЙНО ПРОГРАМИРАНЕ (с ВБ 6.0) – 56 – ПРАКТИЧЕСКА ЧАСТ

© ст. ас. Ст. Анева, 2005 Последна редакция 18 Март 2005 г.

56/59



СЪБИТИЙНО ПРОГРАМИРАНЕ (с ВБ 6.0) – 57 – ПРАКТИЧЕСКА ЧАСТ

© ст. ас. Ст. Анева, 2005 Последна редакция 18 Март 2005 г.

57/59

Програмен код:
Private Sub mnuNew_Click()

rtbDisplay.Text = ""
End Sub
Private Sub mnuExit_Click()

End
End Sub
Private Sub mnuAbout_Click()

MsgBox “I wrote this little editor program”,vbInformation,_
“About FileViewer”
End Sub
Private Sub mnuOpenFile_Click()

Dim strFileName As String
cdlViewer.Filter = "RTF|*.rtf|Text|*.txt|All|*.*"
cdlViewer.ShowOpen
On Error GoTo OpenProblems
strFileName = cdlViewer.filename
rtbDisplay.LoadFile strFileName
lblPath.Caption = strFileName
Exit Sub
OpenProblems:

MsgBox "Can’t open the file, select another.", vbCritical
Exit Sub

End Sub

Private Sub mnuSave_Click()
Dim strFileName As String
cdlViewer.Filter = "RTF|*.rtf|Text|*.txt"
cdlViewer.ShowSave
On Error GoTo SaveProblems
strFileName = cdlViewer.filename
If cdlViewer.FilterIndex = 1 Then

cdlViewer.DefaultExt = “RTF”
rtbDisplay.SaveFile strFileName, rtfRTF

Else
cdlViewer.DefaultExt = “txt”
rtbDisplay.SaveFile strFileName, rtfText

End If
Exit Sub
SaveProblems:

MsgBox "Can’t save the file, try again.", vbCritical
Exit Sub

End Sub

Private Sub mnuBackColor_Click()
cdlViewer.Flags = cdlCCRGBInit
cdlViewer.ShowColor
rtbDisplay.BackColor = cdlViewer.Color

End Sub

Private Sub mnuChangeFont_Click()
cdlViewer.Flags = cdlCFBoth Or cdlCFEffects
cdlViewer.ShowFont
rtbDisplay.SelFontName = cdlViewer.FontName
rtbDisplay.SelFontSize = cdlViewer.FontSize
rtbDisplay.SelBold = cdlViewer.FontBold
rtbDisplay.SelItalic = cdlViewer.FontItalic
rtbDisplay.SelUnderline = cdlViewer.FontUnderline
rtbDisplay.SelColor = cdlViewer.Color

End Sub

Private Sub mnuRichTextBox_Click()
cdlViewer.HelpFile = "VB5def.HLP"
cdlViewer.HelpCommand = cdlHelpKey
cdlViewer.HelpKey = "RichTextbox Control"
cdlViewer.ShowHelp

End Sub


